


USING THE 7854 IN A GPIB CONFIGURATION



Introduction

The Tektronix 7854 Oscilloscope combines high-performance analog capabilities with digital storage, waveform processing, keyboard programmability, and General Purpose Interface Bus (GPIB) interface.

This powerful measurement tool can improve measurement quality; provide push-button solutions to common waveform problems; offer system interface; and dramatically reduce the time from initial display to final calculation.

All the information displayed on the 7854 (except real-time waveforms) can be transferred or received via GPIB interface. This includes stored waveforms and constants, user programs, and other alphanumeric text. Any keystroke command may be remotely issued over the bus, including vertical and horizontal mode select switches. GPIB interface can be used to transfer information for external storage and documentation, for further processing, or to connect with an interactive system that informs the operator of the present status of the system procedure.

As a GPIB system component, the 7854 opens the door to *distributive processing*. The 7854 can be used to perform complete decision-making tasks and send the answer, instead of raw data, over the bus. This allows the GPIB controller to coordinate a more efficient and faster measurement system, and relieves the host computer of processing all the data for a given measurement.

Common bus transfers are essentially transparent to the user, in that the actual data and handshake lines are controlled by a higher level user language. Since single commands will initiate and conduct information transfers, the user need not know the design specifications of the bus or the details of communication transactions.

This application note shows how the 7854 can be interfaced via GPIB with a graphic terminal or a tape unit. Detailed program examples are given. Equipment used in the examples, in addition to the 7854, are the Tektronix GPIB compatible 4052 Graphic Computing System and 4924 Tape Drive. *All examples written in BASIC are also compatible with both the Tektronix 4051 and 4054 Graphic Computing Systems.*

7854 Specifications Summary

The 7854 is a conventional oscilloscope with powerful waveform processing capabilities.

As a conventional oscilloscope, the 7854 offers:

- Dc to 400 MHz bandwidth at 10 mV/div.
- Calibrated sweep rates to 500 ps.
- A choice of more than 30 compatible 7000-Series plug-in units.

With digital storage and waveform processing, the 7854:

- Stores repetitive waveforms up to 400 MHz.
- Reduces noise through signal averaging.
- Stores single shot events (50 μ s/div fastest sweep with 7B87 Plug-in).
- Pretrigger (with 7B87 Plug-in).
- Common waveform parameters at the touch of a key.
- Resolution up to 0.01 div on stored data.
- A choice of 128, 256, 512, or 1024 points per waveform.
- Keystroke programming (up to 1000 lines).
- GPIB interface.




Figure 1. The Tektronix 7854 Oscilloscope shown interfaced with the 4052 Graphic Computing System (left), and with the 4924 Digital Cartridge Tape Drive (right).

Basic Configurations

Two basic GPIB configurations are available with the 7854 Oscilloscope. The first consists of a 7854 interfaced to a mass storage device without any intervention from the controller. For applications requiring permanent storage of 7854-user programs and displayed waveforms, the addition of a GPIB compatible tape drive unit provides an economic GPIB system. No additional software is needed to operate this system.

The second configuration involves the use of a GPIB controller or external computer. For applications that also need full hard copy documentation of further information processing, a GPIB compatible computer and graphics display unit such as the 4052 may be interfaced to the 7854.

Communication Mode

Three GPIB communication modes can be selected: TALK ONLY, LISTEN ONLY, and TALK-LISTEN. In TALK-LISTEN, the end of message terminator can be selected as either EOI ONLY or EOI or LF. This




Figure 2. CRT display of the 7854 shows the ID readout obtained by pressing the ID button.




Figure 3. DIP switches and GPIB connector are located on the rear panel of the 7854.

choice easily permits interfacing with most controllers. The TALK ONLY and LISTEN ONLY modes permit use of a peripheral device, such as a tape drive, for external storage without a controller to coordinate the bus transfers.

Installation and Device Settings

Before making any changes in the rear panel selectors, press the "ID" key on the 7854 mainframe (see Figure 2). The message displayed includes the primary address, the communication mode, and the end of message terminator. If the 7854 is off line (rear selectable) the "OFF LINE" is displayed in place of the mode and message terminator.

If the settings require changing, follow these steps:

- Step 1. Locate the GPIB connector and DIP selector switch on the 7854 upper rear panel. (See Fig. 3.)
- Step 2. Set switches 2 and 3 to the desired transmission mode.*
- Step 3. Set the device's primary address, switches 4-8.
- Step 4. Attach the GPIB cable.

*The selector switches 2 and 3 enable four possible types of transmission. Code 00 and 01 are designed for TALK/LISTEN operation with a controller. Code 00 is for use with TEK controllers such as the 4052. Code 01 allows for differences in the end of message terminator as implemented by other companies. Code 10 is for TALK ONLY, which allows only the transmission of data from the 7854. Code 11 is for LISTEN ONLY, which allows only the reception of data into the 7854.

External Storage to Mag Tape

Manually operating the 4924 Tape Drive, 7854 user programs, waveforms, and numeric constants can be transmitted to magnetic tape without a controller. In this communication, the addresses selected on both 7854 and 4924 are irrelevant.

To insure proper operation with early 4924's, they must be upgraded with the latest firmware. Field modification kits are available. For serial numbers B010100 through B019999, order kit number 050-0860-02; for B020000 through B039999, order kit number 050-0942-01; and for B04000 and above, no changes are necessary.




Figure 4. Displayed keystroke program list from the 7854 Oscilloscope.

Programs

Programs are saved on tape by the following procedure:

- Step 1. Set the 7854 GPIB switches 2 and 3 (rear panel) to Code 10 (TALK ONLY). Press the ID key to verify if desired.
- Step 2. Set the 4924 to off line.
- Step 3. Load a tape into the 4924 Tape Drive.
- Step 4. Advance the tape to the desired file location manually. (REWIND will return to the start of the tape, FORWARD will advance one file at a time, and REVERSE will back up one file.)
- Step 5. Press SAVE on the 7854.
- Step 6. Press LISTEN on the 4924.

The entire program record will be sent and stored on tape. Similarly, a previously stored program can be loaded into the 7854 by the following steps:

- Step 1. Set the GPIB select switch for Code 11 (LISTEN ONLY). Press the ID key to verify.
- Step 2. Load the tape into the 4924.
- Step 3. Advance to the desired tape file manually.
- Step 4. Press PROGRAM ENTRY on the 7854 if in SCOPE, STORED, or BOTH.
- Step 5. Press TALK on the 4924.

When manually operating the 4924 tape drive, the easiest way to locate a file is to press the REWIND button, then use the FORWARD button to advance the tape, one file at a time. To find the first file, press the FORWARD button once. If, for example, the fifth file is desired, press the FORWARD button five times following the REWIND. *All tapes used on the 4924 must be premarked to a specific file length by a 4050-Series computer.* A file length of 10,000 bytes is typically more than enough for most applications. This length will vary depending upon the record transmitted (see appendix for transmission format specifications). If the tape is to be read by a 4050-Series computer, refer to the 4924 Operator Manual for the correct rear panel setting before recording data.

Waveforms and Constants

Waveforms and constants are sent and received with the SENDX and READX commands. The procedure is similar to that for storing programs on magnetic tape. Simply place the desired constant or waveform number in the X-register, then press LISTEN on the 4924 and SENDX on the 7854.

To receive data back into the 7854, press READX, then TALK on the 4924. Remember when using the 4924 Tape Unit that the 7854 must be set to TALK ONLY to send information from the 7854 to the 4924, and LISTEN ONLY to receive information from the 4924.




Figure 5. Storing and retrieving data to and from the 7854 Oscilloscope and 4924 Tape Drive is done using the PROGRAM ENTRY, SAVE, READX, and SENDX on the 7854 and TALK and LISTEN on the 4924.

I/O Procedures Using a Controller

When using the Tektronix 4052 Graphic Computing System as a controller, all keystroke commands can be sent to the 7854 through the GPIB. The 7854 will react as if the commands were issued from the waveform calculator or the measurement keyboard. Commands can be sent to the 7854 from the waveform calculator, the measurement keyboard, the GPIB, or any combination of the above. (When a controller is in use, pressing any key on either 7854 keyboard generates a local rtl message).

To send a command to the 7854 address through the GPIB, send the command mnemonic (or numeric command) in the same way that you would enter the com-

mand from either of the 7854 keyboards, separating each equivalent 7854 keystroke with a space.

Similarly, multiple commands may be grouped as one message by separating each command with a space.

Example 4052 code:

```
PRINT @ 10: "VMDL HMDA AQR RISE"
```

where 10 is the address selected on the 7854.

This code will select the left vertical compartment of the 7854 (VMDL), select the A Horizontal compartment (HMDA), acquire a waveform (AQR), and compute its risetime (RISE). Such a statement may be executed by a BASIC program on the 4052, or may be manually typed in and executed only once as a calculator command.




Figure 6. I/O procedures can be carried out on the 7854 Oscilloscope using the 4052 graphic Computing System as a GPIB compatible controller. Tektronix offers an integrated WP1310 Signal Processing System featuring the 7854 and the 4052. For additional information, see System Specification of the WP1310.

Note: No more than ten commands should be grouped following an I/O command in the same message. The 7854 will buffer up to ten commands following an I/O command, and will execute them on completion of the I/O command. More than ten commands will cause the I/O command to be aborted and all commands following the I/O command to be ignored.

In addition to the measurement commands, the 7854 has commands designed specifically for data transfers through the GPIB. Some of the commands are listed below. (See appendix for complete listing.)

Primary Input/Output Commands

SAVE Outputs the 7854 program memory contents to the GPIB. All program lines are transmitted, from the edit prompt through the end of the program.

SENDX Outputs X-register contents to the GPIB. If the information is a numeric, then that number is sent. But if the information is a waveform number, such as 2 WFM, then the entire waveform record is sent.

READX Inputs a number or waveform from the GPIB and places it in the X-register. Waveform data is stored as 0 WFM.

>TEXT Inputs up to 12 lines of text (40 character maximum for each) from the GPIB for display on the 7854 crt (in STORED, SCOPE, or BOTH modes).

TEXT Outputs a copy of all 16 lines of the currently displayed text to the GPIB.

The SAVE, SENDX, READX, TEXT and >TEXT commands comprise the primary data transfer command set. On the 4052, additional simple BASIC programming statements will be needed to capture incoming data, recall and send data, and handle some of the overhead involved, such as defining and dimensioning variables and responding to a 7854 service request with a serial poll.



Figure 7. Data transferred through the 7854 GPIB interface includes programs, numeric values, waveforms, and text strings. With the 4052, hard copy and tape storage are available.

Data Transfers

Data that can be sent through the GPIB includes programs, numeric values, waveforms, and text strings. The format for all output from the 7854 allows the same data to be sent back into the 7854 without any intermediate processing or loss of information. (The only exception is when the 7854 is set for EOILF as the message terminator in the TALK-LISTEN mode and waveform data is being sent. In this case a carriage return and line feed separate the waveform preamble and curve data. The line feed must be deleted before sending the data back to the 7854. Select EOI when using the 4052.)

Programs

To send a program from the 7854 through the GPIB to a controller for permanent storage on tape or disc, move the edit prompt to the beginning of the program and issue the SAVE command. The program will be sent line-by-line as one record, with the NEXT command and a carriage return appended to the end of each program line. The line numbers are removed from the program as it is sent, and line labels will appear as **LNN a b** where a and b are single digit integers.

These format changes allow the program to be easily re-entered into the 7854 program memory, and the carriage returns allow the controller to transfer the program line-by-line to tape or disc.

The following 4052 BASIC program example (Program 1) will set the 7854 edit prompt at line zero and retrieve and store the program line-by-line on the 4052 magnetic tape.

To retrieve a stored program from tape and send it to the 7854, the 7854 must be in the PROGRAM ENTRY state. Incoming commands

Note: The 7854's device address is 10. The target variable for the incoming data is the ASCII string L\$. The 4052 magnetic tape file is designated by the variable F.

```

100 REM * TRANSFER 7854 PROGRAM TO 4050 TAPE FILE
110 REM * 7854 IS GPIB DEVICE #10
120 REM *
130 REM * STORE PROGRAM ON FILE NUMBER: F
140 REM *
150 ON SRQ THEN 260
160 S=0
170 PRINT @10:"EXECUTE 0 GOTO PROGRAM SAVE"
180 IF S<>208 THEN 180
190 FIND F
200 ON EOI THEN 240
210 INPUT @10:L$
220 PRINT @33:L$
230 GO TO 210
240 PRINT @33:L$
250 END
260 POLL D,S:10
270 RETURN

```

LINE BY LINE COMMENTS:

```

150 Enable SRQ interrupt handler at 260
160 Clear status flag
170 Command 7854 to send entire program
180 Wait for 'SAVE' service request status byte
190 Position 4050 tape to file F
200 Enable EOI interrupt handler at 240
210 Input next line of program from 7854
220 Save line of program on tape
230 Continue until EOI
240 On EOI save last line of program on tape
250 Stop
260 Serial poll 7854: S = status byte
270 Return from SRQ interrupt handler

```

2873 306

Program 1: Data format compatible with Program 2.

will be entered the same as if manually pressed on the waveform calculator. The operator may append a program from tape to an existing program in the 7854 or delete an existing program and load another from tape. The following 4052 BASIC program will delete the existing 7854 program and load in another from the mag tape.

Program 2 functions as the inverse of the previous example. So this will load the same program as the one saved in the previous example.

To append a program to an existing program, remove CLP NEXT from program 2. Position the edit prompt, >, to the location at which you wish to append.

```

100 REM * TRANSFER PROGRAM FROM 4050 TAPE FILE TO 7854
110 REM * 7854 IS GPIB DEVICE #10
120 REM *
130 REM * LOAD PROGRAM FROM FILE NUMBER: F
140 REM *
150 ON SRQ THEN 250
160 S=0
170 PRINT @10:"PROGRAM CLP NEXT"
180 IF S=66 THEN 180
190 FIND F
200 ON EOF (0) THEN 240
210 INPUT @33:L$
220 PRINT @10:L$
230 GO TO 210
240 END
250 POLL D,S:10
260 RETURN

```

LINE BY LINE COMMENTS:

```

150 Enable SRQ interrupt handler at 250
160 Clear status byte
170 Command 7854 to accept a new program
180 Wait for operation complete status byte
190 Position 4050 tape to file F
200 Enable end-of-file interrupt handler at 240
210 Input next line of program from tape file
220 Send line of program to 7854
230 Continue until end-of-file
240 Stop
250 Serial poll 7854; S = status byte
260 Return from SRQ interrupt handler

```

2873-307

Program 2: Data format compatible with Program 1.

SENDX (Numeric Values)

Numeric values (constants) can be transferred from the 7854 over the GPIB by placing the number in the X-register and then executing the SENDX command. The numerical value may be the result of a waveform calculation such as MAX (maximum vertical value), cursor coordinates, or a plug-in's readout (Program 3). See note.

An advantage of sending a calculated result such as MAX is that it relieves the controller or host computer of processing an entire array of waveform data, since the 7854 was designed to make waveform measurements. The result, a single numeric constant can be transferred much faster than an entire array of raw data, making the operation even more efficient.

```

100 REM * TRANSFER CONSTANT (MAX) FROM 7854 TO 4050 VARIABLE
110 REM * 7854 IS GPIB DEVICE #10
120 REM *
130 REM * INPUT CONSTANT INTO VARIABLE: M
140 REM *
150 ON SRQ THEN 210
160 S=0
170 PRINT @10:"MAX SENDX"
180 IF S<>210 THEN 180
190 INPUT @10:M
200 END
210 POLL D,S:10
220 RETURN

```

LINE BY LINE COMMENTS:

```

150 Enable SRQ interrupt handler at line 210
160 Clear status flag
170 Command 7854 to send the 'MAX' value
180 Wait for 'SENDX' service request status byte
190 Input constant (MAX) from 7854 into variable M
200 Stop
210 Serial poll 7854; S = status byte
220 Return from SRQ interrupt handler

```

2873-308

Program 3: Note—The X-register has four digit resolution. To preserve values of greater than four digits, such as from a 7D15 plug-in, use the TEXT command.

READX (Numeric Values)

Numeric values can be transferred back to the 7854 and placed in the X-register by executing the READX command and then sending the number to the 7854 over the GPIB (Program 4).

Alternately, a numeric value may be transferred as individual keystrokes with each command digit separated by a space. This may be more convenient for known constants than for computed variables. For example, to change the points per waveform to 256, send the string "2 5 6 > P/W." The READX may be used, but a simple task like this is performed faster by individual keystroke entries.

```
100 REM *TRANSFER 4050 VARIABLE TO 7854'S X REGISTER
110 REM *7854 IS GPIB DEVICE #10
120 REM *
130 REM *OUTPUT CONSTANT FROM VARIABLE: M
140 REM *
150 ON SRQ THEN 210
160 S=0
170 PRINT @10:"READX"
180 IF S<211 THEN 180
190 PRINT @10:M
200 END
210 POLL D.S:10
220 RETURN
```

LINE BY LINE COMMENTS:

```
150 Enable SRQ interrupt handler at 210
160 Clear status flag
170 Command 7854 to accept data
180 Wait for 'READX' service request status byte
190 Output variable M to 7854
200 Stop
210 Serial poll 7854; S = status byte
220 Return from SRQ interrupt handler
```

2873 309

Program 4

Waveforms

Waveform data is sent as one message (Table 1): the waveform preamble, a separator, and the curve data. The waveform preamble contains information such as points/waveform, volts/division, time/division, and offset voltage. The separator is either a carriage return (Message Terminator Code 00 set to EOI) or a carriage return and line feed (Message Terminator Code 01 set to LF or EOI). The curve data contains the actual vertical ordinate (in divisions ± 5) of each point of the waveform relative to the center crt graticule. The curve data consists of the header (CURVE) followed by a space and then one ASCII coded decimal number for each point of the waveform. Each number is separated by a comma and represents the vertical distance of that point from the center graticule line, measured in graticule divisions (negative if below center). All trailing zeroes are deleted and not sent.

Waveform Preamble: A typical preamble is shown below:

```
WFMPRE ENCDG:ASC,NR.PT:512,PT.FMT:Y,XZERO:0,XINCR:9.766E-06,
XUNIT:S,YZERO:2.704,YMULT:1,YUNIT:V;
```

The abbreviations in the waveform preamble are as follows:

WFMPRE ¹	waveform preamble (header)
ENCDG:ASC ¹	curve data encoded ASCII decimal
NR.PT:(P/W)	number of points/waveform
PT.FMT:Y ¹	point format (curve data in vert. div.)
XZERO:0 ¹	no horizontal offset
XINCR:[10*HSCL/(P/W)]	horizontal increment between points
XUNIT:S	horizontal scale factor units (S=seconds)
YZERO:[-(VSCL*VZR)]	vertical zero offset
YMULT:(VSCL)	vertical scale factor
YUNIT:V	vertical scale factor units (V=volts)

¹Fixed value, cannot be changed.

Curve Data: A typical curve data (partial) is shown below:

```
CURVE 1.3779,1.3777,1.3778,1.3777,...,1.3777,-2.6953,-2.6955,-2.6954,
-2.6955,-2.696,-2.7
```

Table 1

SENDX (Waveforms)

To send stored waveform information from the 7854 through the GPIB, the waveform number (i.e., 3 WFM) must be placed in the X-register. The waveform data can then be sent by giving the SENDX command (Program 5).

The ASCII string format is especially useful for direct storage to mag tape. In this format, the data may be sent back without any intervening processing or reformatting by the controller. The ASCII format is also easy to interpret visually, since it is in a language that can be readily understood by the user.

To facilitate such data manipulation as plotting and further processing, waveform records may easily be input to an array in the 4052. With the 4052 INPUT statement, string variables are converted "on the fly" to array elements as each byte is received. *Any non-numeric character is ignored so the 4052 BASIC program automatically extracts the numeric data from the alphanumeric string.**

Since the data being sent from the 7854 is in a fixed format, the 4052 INPUT statement need only have its variables arranged in the same order (Program 6).

Once the waveform data is transferred to the array, further processing may be performed. Special signal processing ROM packs are available for 4050 Series computers that can provide, for instance, FFT analysis of the waveform to reveal its spectral components and their phase relations. And because these ROM packs operate on an entire array, the processing speed is very fast.

*Other controllers may require you to read in the header information and decode the numeric data. The header CURVE is separated from the first data point of the array by a space, and therefore the header must be extracted from the array.

```
100 REM * TRANSFER WAVEFORM #0 FROM 7854 TO 4050 ASCII STRINGS
120 REM * 7854 IS GPIB DEVICE #10
130 REM *
140 REM * INPUT WAVEFORM PREAMBLE INTO ASCII STRING: P$
150 REM * INPUT WAVEFORM CURVE INTO ASCII STRING: W$
160 REM *
170 DIM P$(200),W$(8200)
180 ON SRQ THEN 250
190 S=0
200 PRINT @10:"0 WFM SENDX"
210 IF S<>210 THEN 210
220 INPUT @10:P$
230 INPUT @10:W$
240 END
250 POLL D,S:10
260 RETURN
```

LINE BY LINE COMMENTS:

```
170 Dimension ASCII strings P$ & W$
180 Enable SRQ interrupt handler at line 250
190 Clear status flag
200 Command 7854 to send waveform #0
210 Wait for 'SENDX' service request status byte
220 Input waveform preamble from 7854 to P$
230 Input waveform curve from 7854 into W$
240 Stop
250 Serial poll 7854; S = status byte
260 Return from SRQ interrupt handler
```

2873-310

Program 5: Data format compatible with Program 7. (Waveforms as ASCII strings)

```
100 REM * TRANSFER WAVEFORM #0 FROM 7854 TO 4050 VARIABLES & ARRAY
110 REM * 7854 IS GPIB DEVICE #10
120 REM *
130 REM * INPUT WAVEFORM PREAMBLE INTO VARIABLES: N,Z1,X,Z2,Y
140 REM * N = POINTS/WAVEFORM
150 REM * Z1 = HORIZONTAL ZERO
160 REM * X = HORIZONTAL INCREMENT BETWEEN POINTS
170 REM * Z2 = VERTICAL ZERO
180 REM * Y = VERTICAL SCALE FACTOR
190 REM * INPUT WAVEFORM CURVE INTO ARRAY: W
200 REM *
210 ON SRQ THEN 300
220 S=0
230 PRINT @10:"0 WFM SENDX"
240 IF S<>210 THEN 240
250 INPUT @10:N,Z1,X,Z2,Y
260 DELETE W
270 DIM W(N)
280 INPUT @10:W
290 END
300 POLL D,S:10
310 RETURN
```

LINE BY LINE COMMENTS:

```
210 Enable SRQ interrupt handler at line 300
220 Clear status flag
230 Command 7854 to send waveform #0
240 Wait for 'SENDX' service request status byte
250 Input waveform preamble arguments into N,Z1,X,Z2,Y
260 Delete Array W
270 Dimension array W to size N (N = Points/Waveform)
280 Input waveform curve into array W
290 Stop
300 Serial poll 7854; S = status byte
310 Return from SRQ interrupt handler
```

2873-311

Program 6: Data format compatible with Program 8. (Waveforms as numeric arrays)

READX (Waveforms)

To copy a waveform back into the 7854, issue the READX command and send the waveform preamble and curve data to the 7854 address. The waveform preamble and curve data must be sent as a single message with no line feed between the two strings. The waveform will become the operational waveform (Ø WFM) in the 7854 and can then be copied to another waveform memory address if desired.

This program requires data formatting in two string variables, the Preamble (P\$) and the Curve (W\$). The first program used with the "SENDX" for waveforms (Program 5) should be used with this READX program (Program 7) to insure the proper data formats.

If the data is not in a suitable format for transmitting to the 7854, the user must reconstruct the Preamble and Curve. If the waveform points are in an array and properly scaled, they may be sent as an array from the 4052 without being converted to an ASCII string.

*When constructing the Preamble it is not necessary to have the exact arrangement of parameters.**

The 7854 will accept the parameters in random fashion provided they agree with the basic format (i.e., NR.PT: 512,). Not all of the preamble must be sent back, but it must contain "WFMPRE" as the first word, "XINCR," "YZERO," and "YMULT" and their respective values to fully specify the waveform. If not transferred, default values are assigned (1 for scale factors and 0 for offset values).

Program 8 constructs a proper waveform preamble from numeric values. These values are converted to string variables and concatenated into a preamble string (P\$). The array (W) is sent without any conversion.

This program's data format is compatible with the second program used with the SENDX command for transferring waveforms (Program 6).

*However, the header CURVE followed by a space must precede the data.

```
100 REM * TRANSFER 4050 ASCII STRINGS TO 7854'S WAVEFORM #0
110 REM * 7854 IS GPIB DEVICE #10
120 REM * OUTPUT WAVEFORM PREAMBLE FROM ASCII STRING: P$
140 REM * OUTPUT WAVEFORM CURVE FROM ASCII STRING: W$
150 REM *
160 ON SRQ THEN 220
170 S=0
180 PRINT @10:"READX"
190 IF S = 211 THEN 190
200 PRINT @10:P$:W$
210 END
220 POLL D.S:10
230 RETURN
```

LINE BY LINE COMMENTS:

```
160 Enable SRQ interrupt handler at 220
170 Clear status flag
180 Command 7854 to accept data
190 Wait for 'READX' service request status byte
200 Output waveform from P$ & W$ to 7854
210 Stop
220 Serial poll 7854: S = status byte
230 Return from SRQ interrupt handler
```

2873-312

Program 7: Data format compatible with Program 5. (Waveforms as ASCII strings)

```
100 REM * TRANSFER 4051 VARIABLES AND ARRAY TO 7854'S WAVEFORM #0
110 REM *
120 REM * OUTPUT WAVEFORM PREAMBLE FROM VARIABLES: N,Z1,X,Z2,Y
130 REM *           N = POINTS/WAVEFORM
140 REM *           Z1 = HORIZONTAL ZERO
150 REM *           X = HORIZONTAL INCREMENT BETWEEN POINTS
160 REM *           Z2 = VERTICAL ZERO
170 REM *           Y = VERTICAL SCALE FACTOR
180 REM * OUTPUT WAVEFORM CURVE FROM ARRAY: W
190 REM *
200 DIM P$(200)
210 ON SRQ THEN 460
220 REM * BUILD WAVEFORM PREAMBLE
230 P$="WFMPRE ENCDG:ASC,NR,PT:"
240 T$=STR(N)
250 P$=P$&T$
260 P$=P$&"",PT.FMT:Y,XZERO:"
270 T$=STR(Z1)
280 P$=P$&T$
290 P$=P$&"",XINCR:"
300 T$=STR(X)
310 P$=P$&T$
320 P$=P$&"",XUNIT:S,YZERO:"
330 T$=STR(Z2)
340 P$=P$&T$
350 P$=P$&"",YMULT:"
360 T$=STR(Y)
370 P$=P$&T$
380 P$=P$&"",YUNIT:V,CURVE "
390 S=0
400 PRINT @10:"READX"
410 IF S > 211 THEN 410
420 REM * THE COMMA(,) BETWEEN P$ AND W IN LINE 440 IS REQUIRED.
430 REM * A SEMICOLON (;) WILL NOT WORK
440 PRINT @10:P$,W
450 END
460 POLL D.S:10
470 RETURN
```

LINE BY LINE COMMENTS:

```
200 Dimension ASCII string P$
210 Enable SRQ interrupt handler at line 460
230 through 380:
    Build waveform preamble in P$
390 Clear status flag
400 Command 7854 to accept data
410 Wait for 'READX' service request status byte
440 Send waveform to 7854
450 Stop
460 Serial poll 7854: S = status byte
470 Return from SRQ interrupt handler
```

2983-313

Program 8: Data format compatible with Program 6. (Waveforms as numeric arrays)

TEXT (Display Message to the 7854 CRT)

Up to 12 lines of text (of up to 40 characters each) may be send to the 7854 for display on the crt along with the waveform(s). This text can include operator prompts, additional information for documenting photographs, or any comments the programmer may wish to display on the 7854 crt.




Figure 8. TEXT message instructing the 7854 Oscilloscope operator to perform a task.

To send text to the 7854, give the >TEXT command, followed by a single string of text with carriage return codes (ASCII decimal 13) inserted at the line breaks (see Program 9). The text display remains until another >TEXT command is given, a crt display command is given (STORED, SCOPE, BOTH, or ID), or the 7854 is put into PROGRAM Mode.

```

100 REM * TRANSFER TEXT FROM 4050 TO 7854 CRT
110 REM * 7854 IS GPIB DEVICE #10
120 REM *
130 REM * BUILD TEXT IN STRING: T$
140 REM *
150 ON SRQ THEN 340
160 DIM T$(12*41)
170 PRINT "INPUT TEXT AS IT IS TO BE DISPLAYED ON 7854"
180 PRINT "┘<SPACE><RETURN> FOR A BLANK LINE"
190 PRINT "<RETURN> ONLY TO END INPUT┘"
200 T$=""
210 C$=CHR(13)
220 FOR I=1 TO 12
230 PRINT "-->";
240 INPUT L$
250 IF LEN(L$)<1 THEN 290
260 T$=T$&L$
270 T$=T$&C$
280 NEXT I
290 S=0
300 PRINT @10:"EXECUTE >TEXT"
310 IF S<213 THEN 310
320 PRINT @10:T$
330 END
340 POLL D.S:10
350 RETURN

```

LINE BY LINE COMMENTS:

```

150 Enable SRQ interrupt handler at line 340
160 Dimension ASCII string T$ (12 lines of 41 characters)
170 through 190:
    Prompt 4050 user to input text
200 Null T$
210 Define C$ as <CARRIAGE RETURN>
220 Loop to accept maximum of 12 lines
230 Issue line input prompt
240 Input next line of text
250 If only <CARRIAGE RETURN> then done inputting
260 Concatenate text onto T$
270 Concatenate a <CARRIAGE RETURN> after each line
280 Continue accepting text
290 Clear status flag
300 Command 7854 to accept text for display
310 Wait for '>TEXT' status byte
320 Send text to 7854
330 Stop
340 Serial poll 7854; S = status byte
350 Return from SRQ interrupt handler

```

2873-314

Program 9

The text to be displayed may include characters not found on the ASCII code chart (see appendix). Table 5 in the appendix lists the additional special characters which may be displayed on the 7854 crt and the codes used to represent them. For example, to display a μ symbol, send ESC followed by CTRL-S.

All 16 lines of text being displayed on the 7854 can be sent back to the controller for editing or inspection by giving the TEXT command. This text includes all text previously sent to the 7854 via the > TEXT command (lines 3 through 14), and all crt readout text (lines 1-2 and 15-16).*

To capture the readout of line 16 of the display, the line must be retrieved before the TEXT command is issued. Otherwise, it will be altered by the TEXT command itself. By using the "SRQ?" command (accessible through the GPIB only), the entire readout of line 16 is returned so that processor status, X and Y registers, and the last executed command may be examined (Program 10). A complete list of GPIB only commands is included in the appendix.

*This crt readout may be either the "SCOPE" or "STORED" displayed characters. When using 7000 Series plug-ins that generate readout of greater resolution than four digits, it is necessary to use the TEXT command while in the "SCOPE" mode to accurately report the readout. This will be required for plug-ins such as the 7D15, 7D13A, 7D12/M2, and the 7D11.

```

100 REM * TRANSFER TEXT FROM 7854 DISPLAY TO 4050
110 REM *
120 ON SRQ THEN 200
130 PAGE
140 PRINT @10:"SRQ?"
150 INPUT @10:S$
160 S=POS(S$,"SRQ ",1)
170 S$=SEG(S$,S+4,LEN(S$)-S-3)
180 S=0
190 PRINT @10:"TEXT"
200 IF S<>212 THEN 200
210 FOR I=1 TO 15
220 INPUT @10:L$
230 PRINT L$
240 NEXT I
250 PRINT S$
260 INPUT @10:L$
270 END
280 POLL D.S:10
290 RETURN

```

LINE BY LINE COMMENTS:

```

120 Enable SRQ interrupt handler at line 200
130 Page 4050 crt screen
140 Use 'SRQ?' querie to get line #16 of 7854's readout
150 Input line #16
160 Find position of substring 'SRQ' in response
170 Remove querie header
180 Clear status flag
190 Command 7854 to send readout
200 Wait for 'TEXT' service request status byte
210 Loop to input first 15 lines of text
220 Input line of text
230 Display line of text on 4050
240 Continue inputting first 15 lines
250 Use 16th line from 'SRQ?' querie
260 Clear last line of text from 7854
270 Stop
280 Serial poll 7854: S = status byte
290 Return from SRQ interrupt handler

```

2873-315

Program 10

Conclusion

All information stored in the 7854 is accessible. With the proper steps, a complete copy of the 7854 crt display (STORED) can be reproduced on the 4052 including additional messages for complete documentation.

By combining the preceding program segments, any I/O sequence for a 4052/7854 (WP1310) system can easily be implemented. Augmented by the graphics and computing capabilities of the 4052, the 7854 Oscilloscope becomes a total waveform measurement system.




Figure 9. All of the stored information in the 7854 can be accessed and a duplicate display can be generated by the 4052.




Figure 10. Combining program segments to implement the desired I/O sequence for a 4052 controller can take full advantage of the 7854 as a total waveform measurement system. (See WP1310 Specifications.)

Appendix

Service Request

The service request function allows the 7854 to keep the GPIB controller informed of its status. This is done by asserting SRQ (Service Request) and then sending a status byte to the controller when polled. The status byte is an eight-bit binary number representing the status of the 7854. If status information is not required for a particular application, all SRQ's can be disabled except the Power-On SRQ asserted on completion of the 7854 self-test (if the 7854 is ON LINE when power is applied). Any service request can be cleared by performing a serial poll of the 7854.

The status bytes sent by the 7854, and their meanings, are listed in Table 2.

When the 7854 is on line, an SRQ is activated when powered up. This SRQ must be cleared via a POLL command from the controller. The returned status byte will indicate whether the 7854 powered up in a normal condition or with an internal error detected by its self-test routine. This service request is not selectable by the controller. If the 7854 is set to the off-line condition before power-up, the SRQ is not activated.

Enable Requests

If IOC is set to on (IOCON), an SRQ is activated after a SAVE, SENDX, READX, TEXT, or >TEXT command is initiated. The controller must respond with a POLL command to begin the I/O transfer.

If OPC is set to on (OPCON), an SRQ is activated after a 7854 programmed STOP or END OF PROGRAM is encountered or at the completion of a command group. Such a group may be a single command or sequence of commands resident in the 7854 temporary command buffer.

If EXR is set to on (EXRON), an SRQ is activated when an execution error occurs with the 7854 user program.

If CER is set to on (CERON), an SRQ is activated when an external command error occurs via the GPIB or when the 7854 is "talked" with nothing to say.

7854 Status Bytes		
Condition	Status Byte 8765 4321	Decimal Equivalent
Normal	000X 0000 ^a	(0.16)
Ext. command group complete ¹	0X00 0010 ^b	(2.66)
End of program ¹	0X00 0010 ^b	(2.66)
Execution error ²	0X10 0010 ^b	(34.98)
External command error ³	0X10 0001 ^b	(33.97)
Power On	010X 0001 ^a	(65.81)
Power on check failure	011X 0011 ^a	(99.115)
RQS command executed ⁴	0101 0000	(80)
SAVE command initiated ⁵	1X01 0000 ^b	(144.208)
SENDX command initiated ⁵	1X01 0010 ^b	(146.210)
READX command initiated ⁵	1X01 0011 ^b	(147.211)
TEXT command initiated ⁵	1X01 0100 ^b	(148.212)
>TEXT command initiated ⁵	1X01 0101 ^b	(149.213)

¹OPCOFF command will disable this service request.
²EXROFF command will disable this service request.
³CEROFF command will disable this service request.
⁴REMOFF command will disable this service request.
⁵IOCOFF command will disable this service request.

^abit 5 true (1) indicates busy condition.
^bbit 7 true (1) indicates active service request.

NOTE

The RQSOFF command will disable all SRQ's except power on. Therefore, for any other service request to function, the RQSON command must be given, along with the command to enable that service request.

Table 2.

Polling

The serial poll allows the GPIB controller to check the status of any or all instruments on the bus. When the 7854 is polled, it sends the appropriate status byte and, if the 7854 was asserting SRQ, clears its service request. The status byte can then be used to initiate corrective action, proceed with the program, or send a message to the operator (Program 11).

```
100 REM * SERIAL POLL SUBROUTINE #1
110 REM *
120 POLL D.S:10
130 RETURN
140 REM * SERIAL POLL SUBROUTINE #2
150 REM *
160 WBYTE @95.63.24.74:
170 RBYTE S
180 WBYTE @95.25:
190 RETURN
```

LINE BY LINE COMMENTS:

```
100 Use 4050 POLL command if 7854 is requesting service
120 S = status byte
130 End of subroutine #1
140 Use 4050 WBYTE and RBYTE commands if 7854 is not
    requesting service
160 95 = Send GPIB UNTalk command (UNT)
    63 = Send GPIB UNListen command (UNL)
    24 = Send GPIB Serial Poll Enable command (SPE)
    74 = Send 7854's talk address
170 Input status byte (S)
180 95 = Send GPIB UNTalk command (UNT)
    25 = Send GPIB Serial Poll Disable command (SPD)
190 End of subroutine #2
```

2873-316

Program 11

GPIB Only Commands: Queries

To receive the response from a query command, use an INPUT statement on the 4052 to a string variable (i.e., Q\$). This INPUT statement should immediately follow the query statement. For example, the 4052 program statements to generate an "SRQ?" and receive the 7854 output would be:

```
PRINT @ 10:"SRQ?"
INPUT @ 10:Q$
```

where 10 is the 7854 GPIB address and Q\$ is the string variable receiving the information (See Table 3.)

Queries

ERR?	Outputs error status; ERR 00 if no error exists or ERR 01 if an error does exist.
ID?	Outputs a copy of the 7854 identification display to the GPIB. (ID TEK/7854.V79.I.XX.YY.ZK where XX is the ROM version, YY is the ROM revision, and Z is 2, 4, or 8, indicating the number of kilobytes of RAM installed.)
SRQ?	Outputs a copy of the bottom line of the calculator display. (SRQ XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX where XXXX... is the bottom line of the display.)

NOTE

Do not send a query (ERR?, ID?, or SRQ?) to the 7854 following an I/O command (SAVE, SENDX, TEXT, >TEXT, or READX) until the I/O command is completed. A query will interrupt the I/O command and cause unpredictable I/O. (Use a serial poll to determine I/O-command status, if necessary. Status byte > 127 indicates I/O in progress.)

Table 3.

Record Lengths for Mag Tape Files

Programs:

The maximum 7854 program will not exceed 29 kilobytes in a 4050 Series format. The length is given by,

$$\text{Length} = (\text{total number of displayed characters} + \text{spaces}) + (\text{total number of lines}) * (5 + 18)$$

The 5 bytes account for the NEXT command and a space which are not displayed. The 18 bytes are added by the 4052 as each line is stored.

Waveforms:

The maximum record length for waveforms varies with the different points per waveform (P/W) values. The required lengths are: for P/W = 1024, 8400 bytes; for P/W = 512, 4300 bytes; for P/W = 256, 2300 bytes; and for P/W = 128, 1200 bytes.

The maximum length is given by,

$$\text{Preamble length} = 150 \text{ bytes}$$

$$\text{Curve length} = (\text{Number of points per waveform}) * 8 + 24$$

Approximate GPIB Transfer Times

Waveform Record	4052 (4054)		4051	
	Array	String	Array	String
128 points	0.83 sec	0.33 sec	2.5 sec	1.0 sec
256 points	1.6 sec	0.6 sec	4.8 sec	1.8 sec
512 points	3.2 sec	1.2 sec	9.6 sec	3.5 sec
1024 points	6.5 sec	2.6 sec	19.4 sec	7.1 sec

Table 4.

Special >TEXT Characters

In addition to the standard ASCII character set of the 4052, the 7854 may display the special characters listed below:

Special Character Codes			
Special Character	ASCII Code	(decimal)	4050 Commands
β (cap beta)	SOH	1	ESC CTRL-A
γ (gama)	STX	2	ESC CTRL-B
η (eta)	EOT	4	ESC CTRL-D
□ (box)	ENQ	5	ESC CTRL-E
▭ (rectangle)	ACK	6	ESC CTRL-F
δ (delta)	BEL	7	ESC CTRL-G
$^{\circ}$ (degree)	BS	8	ESC CTRL-H
α (alpha)	HT	9	ESC CTRL-I
Δ (cap. delta)	SØ	14	ESC CTRL-N
π (pi)	SI	15	ESC CTRL-O
ρ (rho)	DLE	16	ESC CTRL-P
σ (sigma)	DC1	17	ESC CTRL-Q
τ (tau)	DC2	18	ESC CTRL-R
μ (mu or micro)	DC3	19	ESC CTRL-S
ν (nu)	DC4	20	ESC CTRL-T
Ω (cap omega)	SYN	22	ESC CTRL-V
ω (omega)	ETB	23	ESC CTRL-W
√ (square root)	EM	25	ESC CTRL-Y
- (right arrow)	SUB	26	ESC CTRL-Z
- (left arrow)	ESC	27	ESC CTRL-[
↓ (down arrow)	FS	28	ESC CTRL-]
↑ (up arrow)	GS	29	ESC CTRL-^

Table 5.

Reset States

Device Clear

The 7854 responds to a DCL (Device Clear) or SDC (Selected Device Clear) message by executing a STOP command, aborting any input/output or waveform acquisition command in progress, and clearing all SRQ's. If an external input command (READX or >TEXT) is in progress, the partially filled data area is cleared (Ø WFM data points = 0, X-register = 0, or any text input is deleted) and a warning is displayed. If an external output command (SAVE, SENDX, or TEXT) is in progress, it is terminated and a warning is displayed.

Interface Clear

The IFC (Interface Clear) message interrupts any data input or output. If the 7854 is talking, it will continue from that point when it is again addressed as a talker. If the 7854 is listening, it will continue inputting data when it is again addressed as a listener.

Stop Key

If the 7854 is in the talk-listen mode, pressing the 7854 keyboard STOP key will abort an I/O (Input/Output) command (SAVE, TEXT, >TEXT, SENDX, READX) if the 7854 is not yet addressed as a talker or listener. If data transmission has already started, the STOP command is buffered until the I/O command is finished.

If the 7854 is in the TALK ONLY or the LISTEN ONLY mode, the STOP command will abort an I/O command even if data transmission has started.

For further information, contact:

U.S.A., Asia, Australia, Central & South America, Japan

Tektronix, Inc.
P.O. Box 4828
Portland, OR 97208

For additional literature, or the address and phone number of the Tektronix Sales Office nearest you, contact:
Phone: 800/547-1512
Oregon only 800/452-1877
Telex: 910-467-8708
TLX: 15-1754
Cable: TEKTRONIX

Europe, Africa, Middle East


Tektronix Europe B.V.
European Headquarters
Postbox 827
1180 AV Amstelveen
The Netherlands
Phone: (20) 471146
Telex: 18312 - 18328

Canada

Tektronix Canada Inc.
P.O. Box 6500
Barrie, Ontario L4M 4V3
Phone: 705/737-2700

Tektronix sales and service offices around the world:

Argentina, Australia, Austria, Belgium, Bolivia, Brazil, Canada, Chile, Colombia, Costa Rica, Denmark, East Africa, Ecuador, Egypt, El Salvador, Federal Republic of Germany, Finland, France, Greece, Hong Kong, Iceland, India, Indonesia, Ireland, Israel, Italy, Ivory Coast, Japan, Jordan, Korea, Kuwait, Lebanon, Malaysia, Mexico, Morocco, The Netherlands, New Zealand, Norway, Pakistan, Panama, Peru, Philippines, Portugal, Republic of South Africa, Saudi Arabia, Singapore, Spain, Sri Lanka, Sudan, Surinam, Sweden, Switzerland, Syria, Taiwan, Thailand, Turkey, Tunisia, United Kingdom, Uruguay, Venezuela, Zambia, Zimbabwe.

Copyright © 1982 Tektronix, Inc. All rights reserved. Printed in U.S.A. Tektronix products are covered by U.S. and foreign patents, issued and pending. Information in this publication supersedes that in all previously published material. Specification and price change privileges reserved. TEKTRONIX, TEK SCOPE, MOBILE TELE, QUALIMENT and  are registered trademarks. For further information contact Tektronix, Inc., P.O. Box 500, Beaverton, OR 97077. Phone: (503) 644-0161. TWX 910-467-8708. Cable: TEKTRONIX. Subsidiaries and distributors worldwide.

Tektronix
COMMITTED TO EXCELLENCE